

4 用户指南

4 用户指南

文档版本 01
发布日期 2023-12-22



版权所有 © 华为技术有限公司 2023。保留一切权利。

非经本公司书面许可，任何单位和个人不得擅自摘抄、复制本文档内容的部分或全部，并不得以任何形式传播。

商标声明



HUAWEI和其他华为商标均为华为技术有限公司的商标。

本文档提及的其他所有商标或注册商标，由各自的所有人拥有。

注意

您购买的产品、服务或特性等应受华为公司商业合同和条款的约束，本文档中描述的全部或部分产品、服务或特性可能不在您的购买或使用范围之内。除非合同另有约定，华为公司对本文档内容不做任何明示或暗示的声明或保证。

由于产品版本升级或其他原因，本文档内容会不定期进行更新。除非另有约定，本文档仅作为使用指导，本文档中的所有陈述、信息和建议不构成任何明示或暗示的担保。

华为技术有限公司

地址：深圳市龙岗区坂田华为总部办公楼 邮编：518129

网址：<https://www.huawei.com>

客户服务邮箱：support@huawei.com

客户服务电话：4008302118

安全声明

漏洞处理流程

华为公司对产品漏洞管理的规定以“漏洞处理流程”为准，该流程的详细内容请参见如下网址：

<https://www.huawei.com/cn/psirt/vul-response-process>

如企业客户须获取漏洞信息，请参见如下网址：

<https://securitybulletin.huawei.com/enterprise/cn/security-advisory>

目 录

- 1 华为云 Provider 认证..... 1
- 2 弹性云服务器 ECS..... 3
 - 2.1 创建一台 ECS 实例.....3
 - 2.2 添加磁盘.....4
 - 2.3 绑定弹性公网 IP..... 5
- 3 弹性伸缩 AS.....7
- 4 虚拟私有云 VPC..... 11
 - 4.1 典型网络配置.....11
 - 4.2 高可用虚拟 IP VIP..... 13
- 5 NAT 网关..... 16
- 6 对象存储服务 OBS..... 18
 - 6.1 基本操作..... 18
 - 6.2 静态网站托管.....20
- 7 云容器引擎 CCE..... 23
 - 7.1 创建 CCE Cluster..... 23
 - 7.2 创建 CCE Node..... 26
- 8 云数据库 RDS.....28
 - 8.1 创建 MySQL 数据库.....28
 - 8.2 给 RDS 实例绑定 EIP..... 31
 - 8.3 增加只读实例.....33

1 华为云 Provider 认证

华为云Provider采用AK/SK的方式进行认证，支持静态凭据和环境变量两种配置方式。

静态凭据

通过配置provider块中的region，AK/SK等参数进行认证，例如：

```
provider "huaweicloud" {  
  region = "cn-north-1"  
  access_key = "my-access-key"  
  secret_key = "my-secret-key"  
}
```

使用静态凭据的认证方式比较简单，但需要将AK/SK以明文的形式存储在配置文件中，存在密钥泄露的安全隐患。推荐您使用环境变量的方式进行认证。

环境变量

将region，AK/SK等参数设置为环境变量的方式进行认证，例如：

```
$ export HW_REGION_NAME="cn-north-1"  
$ export HW_ACCESS_KEY="my-access-key"  
$ export HW_SECRET_KEY="my-secret-key"
```

环境变量设置后，只需要声明华为云的provider即可。

```
provider "huaweicloud" {}
```

参数说明

表 1-1 Provider 认证参数

参数名称	是否必选	环境变量	参数说明
region	是	HW_REGION_NAME	华为云服务所在的区域，可以从 地区和终端节点 获取。 如果要在不同的region中创建云服务，可以使用alias或者在云服务对应的resource中指定region参数。

参数名称	是否必选	环境变量	参数说明
access_key	是	HW_ACCESS_KEY	用户的密钥ID，查询方法请参见 访问密钥 。
secret_key	是	HW_SECRET_KEY	用户的私有访问密钥，查询方法请参见 访问密钥 。
domain_name	否	HW_DOMAIN_NAME	华为云账号名称，查询方法请参见 API凭证 。
project_name	否	HW_PROJECT_NAME	华为云项目名称，查询方法请参见 API凭证 。
enterprise_project_id	否	HW_ENTERPRISE_PROJECT_ID	企业项目ID，关于企业项目ID的获取及企业项目特性的详细信息，请参见《 企业管理服务用户指南 》。
max_retries	否	HW_MAX_RETRIES	遇到网络传输问题时请求的最大重试次数，默认值为5。

2 弹性云服务器 ECS

[2.1 创建一台ECS实例](#)

[2.2 添加磁盘](#)

[2.3 绑定弹性公网IP](#)

2.1 创建一台 ECS 实例

应用场景

弹性云服务器（Elastic Cloud Server，ECS）是由CPU、内存、操作系统、云硬盘组成的基础的计算组件。弹性云服务器创建成功后，您就可以像使用自己的本地PC或物理服务器一样，在云上使用弹性云服务器。华为云提供了多种类型的弹性云服务器，可以满足不同的使用场景。在创建之前，您需要根据实际的应用场景确认弹性云服务器的规格类型，镜像类型，磁盘种类等参数，并选择合适的网络参数和安全组规则。

相关资源

[huaweicloud_compute_instance](#)

操作步骤

步骤1 使用data source查询可用区，规格，镜像和网络参数。

创建main.tf文件，输入以下内容，并保存在当前的执行目录中。

```
data "huaweicloud_availability_zones" "myaz" {}

data "huaweicloud_compute_flavors" "myflavor" {
  availability_zone = data.huaweicloud_availability_zones.myaz.names[0]
  performance_type = "normal"
  cpu_core_count   = 2
  memory_size      = 4
}

data "huaweicloud_images_image" "myimage" {
  name      = "Ubuntu 18.04 server 64bit"
  most_recent = true
}

data "huaweicloud_vpc_subnet" "mynet" {
```

```
name = "subnet-default"
}

data "huaweicloud_networking_secgroup" "mysecgroup" {
  name = "default"
}
```

步骤2 创建ECS实例并使用随机密码方式登录。

1. 在main.tf文件中增加以下内容。

```
resource "random_password" "password" {
  length      = 16
  special     = true
  override_special = "!!@#$$%*"
}

resource "huaweicloud_compute_instance" "myinstance" {
  name          = "basic"
  admin_pass    = random_password.password.result
  image_id      = data.huaweicloud_images_image.myimage.id
  flavor_id     = data.huaweicloud_compute_flavors.myflavor.ids[0]
  availability_zone = data.huaweicloud_availability_zones.myaz.names[0]
  security_group_ids = [data.huaweicloud_networking_secgroup.mysecgroup.id]

  network {
    uuid = data.huaweicloud_vpc_subnet.mynet.id
  }
}
```

2. 运行 terraform init 初始化环境。

3. 运行 terraform plan 查看资源。

4. 确认资源无误后，运行 terraform apply 开始创建。

5. 运行 terraform show 查看已创建的ECS。

----结束

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/blob/master/examples/ecs/basic/main.tf>

2.2 添加磁盘

应用场景

创建云硬盘并挂载到已创建的弹性云服务器。

相关资源

- [huaweicloud_evs_volume](#)
- [huaweicloud_compute_volume_attach](#)

操作步骤

步骤1 在main.tf文件中增加以下内容。

```
resource "huaweicloud_evs_volume" "myvolume" {
  name          = "myvolume"
  availability_zone = data.huaweicloud_availability_zones.myaz.names[0]
  volume_type   = "SAS"
}
```



```
size      = 10
}

resource "huaweicloud_compute_volume_attach" "attached" {
  instance_id = huaweicloud_compute_instance.myinstance.id
  volume_id   = huaweicloud_evs_volume.myvolume.id
}
```

步骤2 运行 terraform plan 查看资源。

步骤3 确认资源无误后，运行 terraform apply 开始创建。

步骤4 云硬盘挂载至云服务器后，需要登录云服务器初始化云硬盘，即格式化云硬盘，之后云硬盘才可以正常使用。

----结束

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/blob/master/examples/ecs/attached-volume/main.tf>

2.3 绑定弹性公网 IP

应用场景

购买弹性公网IP并绑定给已创建的弹性云服务器。

相关资源

- [huaweicloud_vpc_eip](#)
- [huaweicloud_compute_eip_associate](#)

操作步骤

步骤1 在main.tf文件中增加以下内容。

```
resource "huaweicloud_vpc_eip" "myeip" {
  publicip {
    type = "5_bgp"
  }
  bandwidth {
    name      = "mybandwidth"
    size      = 8
    share_type = "PER"
    charge_mode = "traffic"
  }
}

resource "huaweicloud_compute_eip_associate" "associated" {
  public_ip = huaweicloud_vpc_eip.myeip.address
  instance_id = huaweicloud_compute_instance.myinstance.id
}
```

步骤2 运行 terraform plan 查看资源。

步骤3 确认资源无误后，运行 terraform apply 开始创建。

----结束

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/blob/master/examples/ecs/associated-eip/main.tf>

3 弹性伸缩 AS

应用场景

弹性伸缩服务可根据用户的业务需求，通过策略自动调整其业务的资源。具有自动调整资源、节约成本开支、提高可用性和容错能力的优势。适用以下场景：

- 访问流量较大的论坛网站，业务负载变化难以预测，需要根据实时监控到的云服务器CPU使用率、内存使用率等指标对云服务器数量进行动态调整。
- 电商网站，在进行大型促销活动时，需要定时增加云服务器数量和带宽大小，以保证促销活动顺利进行。
- 视频直播网站，每天14:00~16:00播出热门节目，每天都需要在该时段增加云服务器数量，增大带宽大小，保证业务的平稳运行。

相关资源

- [huaweicloud_as_configuration](#)
- [huaweicloud_as_group](#)
- [huaweicloud_as_policy](#)
- [huaweicloud_ces_alarmrule](#)

操作步骤

步骤1 创建伸缩配置。

创建main.tf文件，输入以下内容，并保存在当前的执行目录中。

```
data "huaweicloud_availability_zones" "myaz" {}

data "huaweicloud_compute_flavors" "myflavor" {
  availability_zone = data.huaweicloud_availability_zones.myaz.names[0]
  performance_type = "normal"
  cpu_core_count   = 2
  memory_size      = 4
}

data "huaweicloud_images_image" "myimage" {
  name      = "Ubuntu 18.04 server 64bit"
  most_recent = true
}

resource "huaweicloud_as_configuration" "my_as_config" {
  scaling_configuration_name = "my_as_config"
```

```
instance_config {
  flavor = data.huaweicloud_compute_flavors.myflavor.ids[0]
  image  = data.huaweicloud_images_image.myimage.id
  key_name = var.my_keypair
  disk {
    size      = 40
    volume_type = "SSD"
    disk_type  = "SYS"
  }
}
```

步骤2 创建弹性伸缩组。

在main.tf文件中增加以下内容。

```
data "huaweicloud_vpc" "vpc_1" {
  name = var.vpc_name
}

data "huaweicloud_vpc_subnet" "subnet_1" {
  name = var.subnet_name
  vpc_id = data.huaweicloud_vpc.vpc_1.id
}

data "huaweicloud_networking_secgroup" "secgroup_1" {
  name = var.secgroup_name
}

resource "huaweicloud_as_group" "my_as_group" {
  scaling_group_name = "my_as_group"
  scaling_configuration_id = huaweicloud_as_configuration.my_as_config.id
  desire_instance_number = 2
  min_instance_number = 0
  max_instance_number = 10
  vpc_id = data.huaweicloud_vpc.vpc_1.id
  delete_publicip = true
  delete_instances = "yes"
  networks {
    id = data.huaweicloud_vpc_subnet.subnet_1.id
  }
  security_groups {
    id = data.huaweicloud_networking_secgroup.secgroup_1.id
  }
  tags = {
    owner = "AutoScaling"
  }
}
```

步骤3 添加扩容策略。

本教程使用告警策略，当平均CPU使用率 $\geq 80\%$ 时自动扩容一个ECS实例，在main.tf文件中增加以下内容。

```
resource "huaweicloud_ces_alarmrule" "scaling_up_rule" {
  alarm_name = "scaling_up_rule"
  metric {
    namespace = "SYS.AS"
    metric_name = "cpu_util"
    dimensions {
      name = "AutoScalingGroup"
      value = huaweicloud_as_group.my_as_group.id
    }
  }
  condition {
    period = 300
    filter = "average"
    comparison_operator = ">="
    value = 80
  }
}
```

```
    unit      = "%"
    count     = 1
  }
  alarm_actions {
    type      = "autoscaling"
    notification_list = []
  }
}
resource "huaweicloud_as_policy" "scaling_up_policy" {
  scaling_policy_name = "scaling_up_policy"
  scaling_policy_type = "ALARM"
  scaling_group_id    = huaweicloud_as_group.my_as_group.id
  alarm_id            = huaweicloud_ces_alarmrule.scaling_up_rule.id
  cool_down_time      = 300
  scaling_policy_action {
    operation      = "ADD"
    instance_number = 1
  }
}
```

步骤4 添加缩容策略。

本教程使用告警策略，当平均CPU使用率 $\leq 20\%$ 时自动缩容一个ECS实例，在main.tf文件中增加以下内容。

```
resource "huaweicloud_ces_alarmrule" "scaling_down_rule" {
  alarm_name = "scaling_down_rule"
  metric {
    namespace = "SYS.AS"
    metric_name = "cpu_util"
    dimensions {
      name = "AutoScalingGroup"
      value = huaweicloud_as_group.my_as_group.id
    }
  }
  condition {
    period      = 300
    filter      = "average"
    comparison_operator = "<="
    value       = 20
    unit        = "%"
    count       = 1
  }
  alarm_actions {
    type      = "autoscaling"
    notification_list = []
  }
}
resource "huaweicloud_as_policy" "scaling_down_policy" {
  scaling_policy_name = "scaling_down_policy"
  scaling_policy_type = "ALARM"
  scaling_group_id    = huaweicloud_as_group.my_as_group.id
  alarm_id            = huaweicloud_ces_alarmrule.scaling_down_rule.id
  cool_down_time      = 300
  scaling_policy_action {
    operation      = "REMOVE"
    instance_number = 1
  }
}
```

步骤5 配置变量。

创建variables.tf文件，输入以下内容，并保存在当前的执行目录中。您可以根据实际情况修改变量值。

```
variable "my_keypair" {
  default = "default"
}
```

```
variable "vpc_name" {  
    default = "vpc-default"  
}  
variable "subnet_name" {  
    default = "subnet-default"  
}  
variable "secgroup_name" {  
    default = "default"  
}
```

步骤6 创建资源。

1. 运行 terraform init 初始化环境。
2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 查看已创建的资源。

----结束

样例代码

https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/auto-scaling/alarm_policy

4 虚拟私有云 VPC

4.1 典型网络配置

4.2 高可用虚拟IP VIP

4.1 典型网络配置

应用场景

在创建VPC之前，您需要根据具体的业务需求规划VPC的数量，子网的数量、IP网段划分和互联互通方式等。详细的网络规划请参考虚拟私有云[最佳实践](#)。

本教程将创建一个VPC用于托管Web应用或网站，其私网网段为192.168.0.0/16，按照业务模块划分3个子网，子网1用于Web层，子网2用于应用层，子网3用于数据层。将服务器划分到不同的安全组中，按需设置访问控制策略，满足高安全场景。

相关资源

- [huaweicloud_vpc](#)
- [huaweicloud_vpc_subnet](#)
- [huaweicloud_networking_secgroup](#)
- [huaweicloud_networking_secgroup_rule](#)

操作步骤

步骤1 创建虚拟私有云和子网。

1. 创建network.tf文件，输入以下内容，并保存在当前的执行目录中。

```
resource "huaweicloud_vpc" "vpc" {
  name = "vpc-web"
  cidr = "192.168.0.0/16"
}
resource "huaweicloud_vpc_subnet" "subnet1" {
  name      = "subnet-web"
  cidr      = "192.168.10.0/24"
  gateway_ip = "192.168.10.1"
  vpc_id    = huaweicloud_vpc.vpc.id
  dns_list  = ["100.125.1.250", "100.125.129.250"]
}
resource "huaweicloud_vpc_subnet" "subnet2" {
```

```
name      = "subnet-app"
cidr      = "192.168.20.0/24"
gateway_ip = "192.168.20.1"
vpc_id    = huaweicloud_vpc.vpc.id
dns_list  = ["100.125.1.250", "100.125.129.250"]
}
resource "huaweicloud_vpc_subnet" "subnet3" {
  name      = "subnet-db"
  cidr      = "192.168.30.0/24"
  gateway_ip = "192.168.30.1"
  vpc_id    = huaweicloud_vpc.vpc.id
  dns_list  = ["100.125.1.250", "100.125.129.250"]
}
```

表 4-1 参数说明

资源名称	参数名称	参数说明
huaweicloud_vpc	name	虚拟私有云名称。 - 取值范围：1-64个字符，支持数字、字母、中文、_(下划线)、-（中划线）、.（点） - 约束：同一个租户下的名称不能重复
	cidr	虚拟私有云下可用子网的范围，必须是cidr格式，例如:192.168.0.0/16
huaweicloud_vpc_subnet	name	子网名称。 - 取值范围：1-64个字符，支持数字、字母、中文、_(下划线)、-（中划线）、.（点）
	cidr	子网的网段。 - 取值范围：必须在vpc对应cidr范围内 - 约束：必须是cidr格式。掩码长度不能大于28
	gateway_ip	子网的网关
	vpc_id	子网所在VPC标识，此处采用引用的形式：huaweicloud_vpc.vpc.id
	dns_list	子网dns服务器地址的集合，不填时，默认为空。内网DNS地址请参见 华为云提供的内网DNS地址是多少？

2. 运行 terraform init 初始化环境。
3. 运行 terraform plan 查看资源。
4. 确认资源无误后，运行 terraform apply 开始创建。
5. 运行 terraform show 查看已创建的VPC和subnet。

步骤2 创建安全组和安全组规则。

1. 在network.tf文件中增加以下内容。

```
resource "huaweicloud_networking_secgroup" "mysecgroup" {
  name      = "secgroup"
  description = "My security group"
  delete_default_rules = true
}
```



```
}  
resource "huaweicloud_networking_secgroup_rule" "secgroup_rule" {  
  direction      = "ingress"  
  ethertype       = "IPv4"  
  protocol        = "tcp"  
  port_range_min  = 22  
  port_range_max  = 22  
  remote_ip_prefix = "0.0.0.0/0"  
  security_group_id = huaweicloud_networking_secgroup.mysecgroup.id  
}
```

2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 查看已创建的安全组和安全组规则。

----结束

样例代码

- <https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/vpc/basic>
- <https://github.com/huaweicloud/terraform-provider-huaweicloud/blob/master/examples/vpc/secgroup/main.tf>

4.2 高可用虚拟 IP VIP

应用场景

虚拟IP主要用在弹性云服务器的主备切换，达到高可用性HA（High Availability）的目的。当主服务器发生故障无法对外提供服务时，动态将虚拟IP切换到备服务器，继续对外提供服务。

相关资源

- [huaweicloud_networking_vip](#)
- [huaweicloud_networking_vip_associate](#)

操作步骤

步骤1 配置网络。

创建main.tf文件，输入以下内容，并保存在当前的执行目录中。

```
resource "huaweicloud_vpc" "vpc_1" {  
  name = var.vpc_name  
  cidr = var.vpc_cidr  
}  
  
resource "huaweicloud_vpc_subnet" "subnet_1" {  
  vpc_id      = huaweicloud_vpc.vpc_1.id  
  name        = var.subnet_name  
  cidr        = var.subnet_cidr  
  gateway_ip  = var.subnet_gateway  
  primary_dns = var.primary_dns  
}
```

步骤2 创建两台ECS实例。

在main.tf文件中增加以下内容。

```
data "huaweicloud_availability_zones" "myaz" {}

data "huaweicloud_compute_flavors" "myflavor" {
  availability_zone = data.huaweicloud_availability_zones.myaz.names[0]
  performance_type = "normal"
  cpu_core_count   = 2
  memory_size      = 4
}

data "huaweicloud_images_image" "myimage" {
  name      = "Ubuntu 18.04 server 64bit"
  most_recent = true
}

data "huaweicloud_networking_secgroup" "mysecgroup" {
  name = "default"
}

resource "huaweicloud_compute_instance" "mycompute" {
  name          = "mycompute_${count.index}"
  image_id      = data.huaweicloud_images_image.myimage.id
  flavor_id     = data.huaweicloud_compute_flavors.myflavor.ids[0]
  availability_zone = data.huaweicloud_availability_zones.myaz.names[0]
  security_group_ids = [data.huaweicloud_networking_secgroup.mysecgroup.id]
  network {
    uuid = huaweicloud_vpc_subnet.subnet_1.id
  }
  count = 2
}
```

步骤3 申请虚拟IP地址并绑定ECS服务器对应的端口。

在main.tf文件中增加以下内容。

```
resource "huaweicloud_networking_vip" "vip_1" {
  network_id = huaweicloud_vpc_subnet.subnet_1.id
}

# associate ports to the vip
resource "huaweicloud_networking_vip_associate" "vip_associated" {
  vip_id = huaweicloud_networking_vip.vip_1.id
  port_ids = [
    huaweicloud_compute_instance.mycompute[0].network.0.port,
    huaweicloud_compute_instance.mycompute[1].network.0.port
  ]
}
```

步骤4 配置变量。

创建variables.tf文件，输入以下内容，并保存在当前的执行目录中。您可以根据实际情况修改变量值。

```
variable "vpc_name" {
  default = "vpc-basic"
}
variable "vpc_cidr" {
  default = "172.16.0.0/16"
}
variable "subnet_name" {
  default = "subent-basic"
}
variable "subnet_cidr" {
  default = "172.16.10.0/24"
}
variable "subnet_gateway" {
  default = "172.16.10.1"
}
variable "primary_dns" {
  default = "100.125.1.250"
}
```

步骤5 创建资源。

1. 运行 terraform init 初始化环境。
2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 查看已创建的资源。

----结束

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/vpc/vip>

5 NAT 网关

应用场景

当多个云主机在没有绑定弹性公网IP的情况下需要访问公网，为了节省弹性公网IP资源并且避免云主机IP直接暴露在公网上，可以通过公网NAT网关共享弹性公网IP的方式访问公网，实现无弹性公网IP的云主机访问公网。

相关资源

[huaweicloud_vpc_eip](#)

[huaweicloud_nat_gateway](#)

[huaweicloud_nat_snat_rule](#)

操作步骤

步骤1 申请弹性公网IP。

创建main.tf文件，输入以下内容，并保存在当前的执行目录中。

```
resource "huaweicloud_vpc_eip" "eip_1" {  
  publicip {  
    type = "5_bgp"  
  }  
  bandwidth {  
    name      = "test"  
    size      = 5  
    share_type = "PER"  
    charge_mode = "traffic"  
  }  
}
```

步骤2 申请NAT网关并配置SNAT规则。

在main.tf文件中增加以下内容。

```
data "huaweicloud_vpc" "vpc_1" {  
  name = "vpc-default"  
}  
  
data "huaweicloud_vpc_subnet" "subnet_1" {  
  name      = "subnet-default"  
  vpc_id = data.huaweicloud_vpc.vpc_1.id  
}
```

```
resource "huaweicloud_nat_gateway" "nat_1" {
  name      = "nat-gateway-basic"
  description = "test for terraform examples"
  spec      = "1"
  vpc_id    = data.huaweicloud_vpc.vpc_1.id
  subnet_id = data.huaweicloud_vpc_subnet.subnet_1.id
}

resource "huaweicloud_nat_snat_rule" "snat_1" {
  floating_ip_id = huaweicloud_vpc_eip.eip_1.id
  nat_gateway_id = huaweicloud_nat_gateway.nat_1.id
  network_id     = data.huaweicloud_vpc_subnet.subnet_1.id
}
```

表 5-1 参数说明

资源名称	参数名称	参数说明
huaweicloud_nat_gateway	name	NAT网关的名字，支持数字、字母、_（下划线）、-（中划线）、中文。
	description	NAT网关的描述。
	spec	NAT网关的规格。取值为： <ul style="list-style-type: none">“1”：小型，SNAT最大连接数10000“2”：中型，SNAT最大连接数50000“3”：大型，SNAT最大连接数200000“4”：超大型，SNAT最大连接数1000000
	vpc_id	VPC的ID。
	subnet_id	子网的网络ID。
huaweicloud_nat_snat_rule	floating_ip_id	弹性公网IP的id，多个弹性公网IP使用逗号分隔。 <ul style="list-style-type: none">约束：弹性公网IP的id个数不能超过20个。
	nat_gateway_id	NAT网关的ID。
	network_id	规则使用的网络ID。

步骤3 创建资源。

1. 运行 terraform init 初始化环境。
2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。

----结束

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/nat/snate-basic>

6 对象存储服务 OBS

- 6.1 基本操作
- 6.2 静态网站托管

6.1 基本操作

应用场景

对象存储服务OBS（Object Storage Service）提供海量、安全、高可靠、低成本的数据存储能力，可供用户存储任意类型和大小数据。适合企业备份/归档、视频点播、视频监控等多种数据存储场景。

相关资源

- [huaweicloud_obs_bucket](#)
- [huaweicloud_obs_bucket_object](#)

操作步骤

步骤1 创建OBS存储桶。

1. 创建main.tf文件，输入以下内容，并保存在当前的执行目录中。

```
resource "huaweicloud_obs_bucket" "myexample" {  
  bucket = "myexample-bucket"  
  acl    = "private"  
  tags = {  
    type = "bucket"  
    env  = "Test"  
  }  
}
```

2. 运行 terraform init 进行初始化。
3. 运行 terraform plan 查看资源。
4. 确认资源无误后，运行 terraform apply 开始创建。
5. 运行 terraform show 可以查看已创建的OBS桶。

步骤2 上传对象。

1. 支持通过数据流和源文件路径上传对象，在main.tf文件中增加以下内容。

```
# 通过数据流上传对象
resource "huaweicloud_obs_bucket_object" "myobject1" {
  bucket      = huaweicloud_obs_bucket.myexample.bucket
  key         = "myobject1"
  content     = "content of myobject1"
  content_type = "application/xml"
}
# 通过源文件路径上传对象
resource "huaweicloud_obs_bucket_object" "myobject2" {
  bucket = huaweicloud_obs_bucket.myexample.bucket
  key    = "myobject2"
  source = "hello.txt"
}
# 通过源文件路径上传对象并进行服务端加密
resource "huaweicloud_obs_bucket_object" "myobject3" {
  bucket      = huaweicloud_obs_bucket.myexample.bucket
  key         = "myobject3"
  source      = "hello.txt"
  encryption = true
}
```
2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 可以查看已上传的对象。

----结束

表 6-1 参数说明

资源名称	参数名称	参数说明
huaweicloud_obs_bucket	bucket	(必填) OBS存储桶名称 <ul style="list-style-type: none">约束：在OBS中必须是全局唯一的，包含3到63个字符，允许小写字母、数字、连字符(-)和点号(.)，不能以句点(.)或连字符(-)开始或结束，不能包含两个连续的句点(.)或相邻的句点(.)和连字符(-)，不能是IP地址。
	acl	(可选) OBS存储桶访问控制策略 <ul style="list-style-type: none">取值范围： private：默认策略，除桶ACL授权外的其他用户无桶的访问权限。 public-read：任何用户都可以对桶内对象进行读操作。 public-read-write：任何用户都可以对桶内对象进行读/写/删除操作。
	tags	(可选) 桶标签
huaweicloud_obs_bucket_object	bucket	(必填) 存储桶名称
	key	(必填) 对象名称
	source	(可选) 待上传对象的源文件路径
	content	(可选) 待上传对象的数据流
	content_type	(可选) 待上传对象的MIME类型
	encryption	(可选) 是否开启桶的默认服务端加密

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/blob/master/examples/obs/basic/main.tf>

6.2 静态网站托管

应用场景

OBS服务允许在桶上托管静态网站，支持索引页、错误页展示和页面跳转。您可以将静态网站文件上传至OBS的桶中，并对这些文件赋予匿名用户可读权限，然后将该桶配置成静态网站托管模式，就可以实现在OBS上托管静态网站了。

相关资源

- [huaweicloud_obs_bucket](#)
- [huaweicloud_obs_bucket_object](#)
- [huaweicloud_obs_bucket_policy](#)

操作步骤

步骤1 创建OBS存储桶，并配置静态网站托管。

1. 创建main.tf文件，输入以下内容，并保存在当前的执行目录中。

```
resource "huaweicloud_obs_bucket" "mywebsite" {  
  bucket = "mywebsite"  
  website {  
    index_document = "index.html"  
    error_document = "error.html"  
  }  
}
```
2. 运行 terraform init 进行初始化。
3. 运行 terraform plan 查看资源。
4. 确认资源无误后，运行 terraform apply 开始创建。
5. 运行 terraform show 可以查看已创建的OBS桶。

步骤2 配置桶策略，允许匿名用户访问对象。

在main.tf文件中，输入以下内容：

```
# granting the Read-Only permission to anonymous users  
resource "huaweicloud_obs_bucket_policy" "policy" {  
  bucket = huaweicloud_obs_bucket.mywebsite.bucket  
  policy = <<POLICY  
{  
  "Statement": [  
    {  
      "Sid": "AddPerm",  
      "Effect": "Allow",  
      "Principal": {"ID": "*"},  
      "Action": ["GetObject"],  
      "Resource": "mywebsite/*"  
    }  
  ]  
}
```



```
}  
POLICY  
}
```

步骤3 上传静态网站文件。

1. 在当前目录下编辑index.html和error.html文件。
2. 在main.tf文件，输入以下内容，将文件上传至OBS桶。

```
# put index.html  
resource "huaweicloud_obs_bucket_object" "index" {  
  bucket = huaweicloud_obs_bucket.mywebsite.bucket  
  key    = "index.html"  
  source = "index.html"  
}  
# put error.html  
resource "huaweicloud_obs_bucket_object" "error" {  
  bucket = huaweicloud_obs_bucket.mywebsite.bucket  
  key    = "error.html"  
  source = "error.html"  
}
```
3. 运行 terraform plan 查看资源。
4. 确认资源无误后，运行 terraform apply 开始上传。

步骤4 结果验证。

通过浏览器访问 <https://mywebsite.obs-website.cn-north-4.myhuaweicloud.com>，即可访问index.html。*这里mywebsite为OBS桶名称，cn-north-4为存储桶所属的region。*

----结束

表 6-2 参数说明

资源名称	参数名称		参数说明
huaweicloud_obs_bucket	bucket		(必填) OBS存储桶名称 <ul style="list-style-type: none">约束：在OBS中必须是全局唯一的，包含3到63个字符，允许小写字母、数字、连字符(-)和点号(.)，不能以句点(.)或连字符(-)开始或结束，不能包含两个连续的句点(.)或相邻的句点(.)和连字符(-)，不能是IP地址。
	website	index_document	(必填) 访问静态网站时返回的索引页面，即首页。
		error_document	(可选) 当访问错误的静态网站路径时，返回的404错误页面。
		routing_rules	(可选) 静态网站重定向规则
huaweicloud_obs_bucket_policy	bucket		(必填) 存储桶名称
	policy_format		(可选) 策略格式，支持obs和s3两种类型格式，默认为obs类型
	policy		(必填) 策略内容，请参考 Policy格式

资源名称	参数名称	参数说明
huaweicloud_obs_bucket_object	bucket	(必填) 存储桶名称
	key	(必填) 对象名称
	source	(可选) 待上传对象的源文件路径

后续操作

您可以将自己的域名与OBS桶访问域名进行绑定，可以实现通过自定义域名访问存储在OBS中的文件。同时也可以配置CDN服务实现加速功能。请参考[使用自定义域名托管静态网站](#)。

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/obs/website>

7 云容器引擎 CCE

云容器引擎（Cloud Container Engine，简称CCE）提供高度可扩展的、高性能的企业级Kubernetes集群，支持运行Docker容器。借助云容器引擎，您可以在华为云上轻松部署、管理和扩展容器化应用程序。本教程将指导您如何通过Terraform脚本创建一个CCE Cluster和 Node

[7.1 创建CCE Cluster](#)

[7.2 创建CCE Node](#)

7.1 创建 CCE Cluster

相关资源

- [huaweicloud_vpc](#)
- [huaweicloud_vpc_subnet](#)
- [huaweicloud_vpc_eip](#)
- [huaweicloud_cce_cluster](#)

操作步骤

步骤1 创建虚拟私有云和子网。请参见[4.1 典型网络配置](#)。

1. 创建cce.tf文件，输入以下内容，并保存在当前的执行目录中。

```
resource "huaweicloud_vpc" "myvpc" {
  name = "myvpc"
  cidr = "192.168.0.0/16"
}

resource "huaweicloud_vpc_subnet" "mysubnet" {
  name      = "mysubnet"
  cidr      = "192.168.0.0/16"
  gateway_ip = "192.168.0.1"

  //dns is required for cce node installing
  primary_dns   = "100.125.1.250"
  secondary_dns = "100.125.21.250"
  vpc_id        = huaweicloud_vpc.myvpc.id
}
```

2. 运行 terraform init 初始化环境。

3. 运行 terraform plan 查看资源。
4. 确认资源无误后，运行 terraform apply 开始创建。
5. 运行 terraform show 查看已创建的VPC和subnet。

步骤2 创建弹性公网IP，若集群不使用公网可跳过此步骤。

1. 在cce.tf文件中增加以下内容。

```
resource "huaweicloud_vpc_eip" "myeip" {  
  publicip {  
    type = "5_bgp"  
  }  
  bandwidth {  
    name      = "mybandwidth"  
    size      = 8  
    share_type = "PER"  
    charge_mode = "traffic"  
  }  
}
```
2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 查看已创建的弹性公网IP。

步骤3 创建CCE Cluster。

1. 在cce.tf文件中增加以下内容。

```
resource "huaweicloud_cce_cluster" "mycce" {  
  name          = "mycce"  
  flavor_id     = "cce.s1.small"  
  vpc_id        = huaweicloud_vpc.myvpc.id  
  subnet_id     = huaweicloud_vpc_subnet.mysubnet.id  
  container_network_type = "overlay_l2"  
  eip           = huaweicloud_vpc_eip.myeip.address // 若不使用弹性公网ip，忽略此行  
}
```
2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 查看已创建的CCE Cluster。

----结束

表 7-1 参数说明

资源名称	参数名称	参数说明
huaweicloud_cce_cluster	name	(必填) 集群名称 <ul style="list-style-type: none">取值范围：以小写字母开头，由小写字母、数字、中划线(-)组成，长度范围4-128位，且不能以中划线(-)结尾。约束：同一个租户下的名称不能重复

资源名称	参数名称	参数说明
	flavor_id	(必填) 集群规格 <ul style="list-style-type: none">取值范围：<ul style="list-style-type: none">cce.s1.small: 小规模单控制节点混合集群（最大50节点）cce.s1.medium: 中等规模单控制节点混合集群（最大200节点）cce.s2.small: 小规模多控制节点混合集群（最大50节点）cce.s2.medium: 中等规模多控制节点混合集群（最大200节点）cce.s2.large: 大规模多控制节点混合集群（最大1000节点）cce.s2.xlarge: 超大规模多控制节点混合集群（最大2000节点）约束：集群创建完成后规格不可再变更，请按实际业务需求进行选择
	vpc_id	(必填) 用于创建控制节点的VPC的ID
	subnet_id	(必填) 用于创建控制节点的subnet的网络ID
	container_network_type	(必填) 容器网络类型 <ul style="list-style-type: none">取值范围：<ul style="list-style-type: none">overlay_l2: 通过OVS（OpenVSwitch）为容器构建的overlay_l2网络。underlay_ipvlan: 裸金属服务器使用ipvlan构建的Underlay的l2网络。vpc-router: 使用ipvlan和自定义VPC路由为容器构建的Underlay的l2网络。eni: Yangtse网络，深度整合VPC原生ENI弹性网卡能力，采用VPC网段分配容器地址，支持ELB直通容器，享有高性能，创建CCE Turbo集群（公测中）时指定。
	eip	(可选) 弹性公网IP

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/cce/basic>

7.2 创建 CCE Node

相关资源

- [huaweicloud_availability_zones](#)
- [huaweicloud_compute_keypair](#)
- [huaweicloud_cce_cluster](#)
- [huaweicloud_cce_node](#)

操作步骤

步骤1 创建CCE Cluster，详细步骤参考[7.1 创建CCE Cluster](#)。

步骤2 创建CCE Node。

1. 在 [7.1 创建CCE Cluster](#) 中创建的cce.tf文件中增加以下内容。

```
data "huaweicloud_availability_zones" "myaz" {}

resource "huaweicloud_compute_keypair" "mykeypair" {
  name = "mykeypair"
}

resource "huaweicloud_cce_node" "mynode" {
  cluster_id = huaweicloud_cce_cluster.mycce.id
  name       = "mynode"
  flavor_id  = "t6.large.2"
  availability_zone = data.huaweicloud_availability_zones.myaz.names[0]
  key_pair   = huaweicloud_compute_keypair.mykeypair.name

  root_volume {
    size       = 40
    volumetype = "SAS"
  }
  data_volumes {
    size       = 100
    volumetype = "SAS"
  }
}
```

2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 查看已创建的CCE Node。

----结束

表 7-2 参数说明

资源名称	参数名称	参数说明
huaweicloud_cce_node	cluster_id	(必填) 集群 ID
	name	(可选) 节点名称 <ul style="list-style-type: none">取值范围：以小写字母开头，由小写字母、数字、中划线(-)组成，长度范围1-56位，且不能以中划线(-)结尾。

资源名称	参数名称		参数说明
	flavor_id		(必填) 节点的规格
	availability_zone		(必填) 节点所在的可用区名 <ul style="list-style-type: none">约束：底层实际存在，位于该用户物理可用区组之内的可用区。
	key_pair		(可选) 选择密钥对方式登录时的密钥对名称 <ul style="list-style-type: none">约束：密钥对(key_pair)和密码(password)登录方式二者必选其一。
	root_volume	size	(必填) 磁盘大小，单位为GB <ul style="list-style-type: none">约束：系统盘取值范围：40~1024
		volume_type	(必填) 磁盘类型 <ul style="list-style-type: none">取值范围：<ul style="list-style-type: none">SATA：普通IO，是指由SATA存储提供资源的磁盘类型。SAS：高IO，是指由SAS存储提供资源的磁盘类型。SSD：超高IO，是指由SSD存储提供资源的磁盘类型。
	data_volume	size	(必填) 磁盘大小，单位为GB <ul style="list-style-type: none">约束：数据盘取值范围：100~32768
		volume_type	(必填) 磁盘类型 <ul style="list-style-type: none">取值范围：<ul style="list-style-type: none">SATA：普通IO，是指由SATA存储提供资源的磁盘类型。SAS：高IO，是指由SAS存储提供资源的磁盘类型。SSD：超高IO，是指由SSD存储提供资源的磁盘类型。

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/cce/basic>

8 云数据库 RDS

云数据库RDS（Relational Database Service，简称RDS）是一种基于云计算平台的即开即用、稳定可靠、弹性伸缩、便捷管理的在线云数据库服务。

[8.1 创建MySQL数据库](#)

[8.2 给RDS实例绑定EIP](#)

[8.3 增加只读实例](#)

8.1 创建 MySQL 数据库

应用场景

MySQL是目前最受欢迎的开源数据库之一，其性能卓越，搭配LAMP（Linux + Apache + MySQL + Perl/PHP/Python），成为WEB开发的高效解决方案。本教程将指导您如何通过Terraform脚本创建一个RDS for MySQL实例。

相关资源

[huaweicloud_rds_instance](#)

操作步骤

步骤1 规划并创建虚拟私有云、子网及安全组。

1. 若需要创建新的网络配置，请参考[4.1 典型网络配置](#)。
2. 若使用已创建的网络配置，需要使用data source获取对应的ID，参考样例如下。

```
data "huaweicloud_vpc" "myvpc" {
  name = var.vpc_name
}
data "huaweicloud_vpc_subnet" "mysubnet" {
  vpc_id = data.huaweicloud_vpc.myvpc.id
  name   = var.subnet_name
}
data "huaweicloud_networking_secgroup" "mysecgroup" {
  name = var.secgroupp_name
}
```

步骤2 创建RDS for MySQL实例。

示例一 使用新建网络配置创建RDS实例，并使用随机密码。


```
data "huaweicloud_availability_zones" "myaz" {}

resource "random_password" "mypassword" {
  length      = 12
  special     = true
  override_special = "!@#%^*-_+= "
}

resource "huaweicloud_rds_instance" "myinstance" {
  name          = "mysql_instance"
  flavor        = "rds.mysql.c2.large.ha"
  ha_replication_mode = "async"
  vpc_id        = huaweicloud_vpc.myvpc.id
  subnet_id     = huaweicloud_vpc_subnet.mysubnet.id
  security_group_id = huaweicloud_networking_secgroup.mysecgroup.id
  availability_zone = [
    data.huaweicloud_availability_zones.myaz.names[0],
    data.huaweicloud_availability_zones.myaz.names[1]
  ]
  db {
    type      = "MySQL"
    version   = "8.0"
    password  = random_password.mypassword.result
  }
  volume {
    type = "ULTRAHIGH"
    size = 40
  }
}
```

示例二 使用已有网络配置创建RDS实例

```
data "huaweicloud_availability_zones" "myaz" {}

resource "huaweicloud_rds_instance" "myinstance" {
  name          = "mysql_instance"
  flavor        = "rds.mysql.c2.large.ha"
  ha_replication_mode = "async"
  vpc_id        = data.huaweicloud_vpc.myvpc.id
  subnet_id     = data.huaweicloud_vpc_subnet.mysubnet.id
  security_group_id = data.huaweicloud_networking_secgroup.mysecgroup.id
  availability_zone = [
    data.huaweicloud_availability_zones.myaz.names[0],
    data.huaweicloud_availability_zones.myaz.names[1]
  ]
  db {
    type      = "MySQL"
    version   = "8.0"
    password  = var.rds_password
  }
  volume {
    type = "ULTRAHIGH"
    size = 40
  }
}
```

步骤3 配置变量。

创建variables.tf文件，输入以下内容，并保存在当前的执行目录中。您可以根据实际情况修改变量值。

```
variable "vpc_name" {
  default = "vpc-basic"
}
variable "vpc_cidr" {
  default = "172.16.0.0/16"
}
variable "subnet_name" {
  default = "subent-basic"
}
variable "subnet_cidr" {
```

```
default = "172.16.10.0/24"
}
variable "subnet_gateway" {
  default = "172.16.10.1"
}
variable "primary_dns" {
  default = "100.125.1.250"
}
```

步骤4 创建资源。

1. 运行 terraform init 进行初始化。
2. 运行 terraform plan 查看资源。
3. 确认资源无误后，运行 terraform apply 开始创建。
4. 运行 terraform show 可以查看已创建的RDS实例信息。

----结束

表 8-1 参数说明

资源名称	参数名称		参数说明
huaweicloud_rds_instance	name		(必填) 数据库实例名称，同一租户下，同类型的数据库实例名称可以重名 <ul style="list-style-type: none">● 约束：4~64个字符之间，必须以字母开头，区分大小写，可以包含字母、数字、中划线或者下划线，不能包含其他的特殊字符
	flavor		(必填) 数据库实例规格，本教程使用 "rds.mysql.c2.large.ha"，实例规格信息也可以通过 huaweicloud_rds_flavors 查询。
	ha_replication_mode		(可选) 备机同步参数。MySQL的取值为 "async" 或 "semisync"。
	availability_zone		(必填) 实例所在可用区，对于主备实例支持多个可用区，请参见 地区和终端节点 。
	vpc_id		(必填) 实例所属的VPC ID
	subnet_id		(必填) 实例所属的子网网络ID
	security_group_id		(必填) 实例所属的安全组ID
	db	type	(必填) 数据库引擎类型 <ul style="list-style-type: none">● 取值范围：MySQL、PostgreSQL、SQLServer
		version	(必填) 数据库引擎版本，MySQL引擎支持5.6、5.7、8.0版本。
		password	(必填) 数据库密码 由大小写字母、数字和特殊符号~!@#%^*_-=+?组成，长度8~32个字符。 建议您输入高强度密码，以提高安全性，防止出现密码被暴力破解等安全风险。

资源名称	参数名称		参数说明
		port	(可选) 数据库端口 <ul style="list-style-type: none">约束: MySQL数据库端口设置范围为1024~65535 (其中12017和33071被RDS系统占用不可设置), 默认3306;
		type	(必填) 数据库实例磁盘类型 <ul style="list-style-type: none">取值范围: ULTRAHIGH, 表示SSD; ULTRAHIGHPRO, 表示SSD尊享版, 仅支持超高性能型尊享版
		size	(必填) 数据库实例磁盘空间大小 <ul style="list-style-type: none">取值范围: 40GB~4000GB, 必须为10的整数倍

样例代码

- <https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/rds/mysql>
- <https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/rds/mysql-with-network>

8.2 给 RDS 实例绑定 EIP

应用场景

云数据库RDS实例创建成功后, 可以绑定弹性公网IP, 来实现通过公共网络访问数据库实例。以下教程将指导您如何通过Terraform脚本给RDS实例绑定和解绑弹性公网IP。

实例在创建、变更规格、恢复、重启或冻结状态下不能绑定和解绑弹性公网IP。

相关资源

- [huaweicloud_rds_instance](#)
- [huaweicloud_vpc_eip](#)
- [huaweicloud_vpc_eip_associate](#)

操作步骤

步骤1 创建MySQL数据库, 请参考[8.1 创建MySQL数据库](#)。

步骤2 增加安全组规则, 允许指定的网络访问RDS实例的端口。

```
resource "huaweicloud_networking_secgroup_rule" "allow_rds" {
  direction      = "ingress"
  ethertype      = "IPv4"
  protocol        = "tcp"
  port_range_min = 3306
  port_range_max = 3306
  remote_ip_prefix = var.allow_cidr
  security_group_id = huaweicloud_networking_secgroup.mysecgroup.id
}
```

步骤3 创建EIP并与RDS实例内网IP进行绑定。

```
# 创建EIP
resource "huaweicloud_vpc_eip" "myeip" {
  publicip {
    type = "5_bgp"
  }
  bandwidth {
    name      = "test"
    size      = 5
    share_type = "PER"
    charge_mode = "traffic"
  }
}
# 查询RDS实例对应的内网网络端口
data "huaweicloud_networking_port" "rds_port" {
  network_id = huaweicloud_vpc_subnet.mysubnet.id
  fixed_ip   = huaweicloud_rds_instance.myinstance.private_ips[0]
}
# 绑定EIP
resource "huaweicloud_vpc_eip_associate" "associated" {
  public_ip = huaweicloud_vpc_eip.myeip.address
  port_id   = data.huaweicloud_networking_port.rds_port.id
}
```

1. 运行 terraform plan 查看资源。
2. 确认资源无误后，运行 terraform apply 开始创建。
3. 运行 terraform show 可以查看已创建的EIP绑定信息。

----结束

表 8-2 参数说明

资源名称	参数名称		参数说明
huaweicloud_vpc_eip	public ip	type	(必填) IP类型，当前仅 5_bgp 可选
	band width	name	(可选) 带宽配置名称
		size	(可选) IP带宽，可选值为1~300Mbit/s
		share_type	(必填) 设置此IP为共享带宽或独享带宽
huaweicloud_networking_port	fixed_ip		(必填) RDS实例的内网IP地址
	network_id		(可选) RDS实例所属的子网网络ID
huaweicloud_vpc_eip_associate	public_ip		(必填) EIP的IP地址
	port_id		(必填) RDS实例对应的Port的ID

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/rds/mysql-with-eip>

8.3 增加只读实例

应用场景

在对数据库有少量写请求，但有大量读请求的应用场景下，单个实例可能无法抵抗读取压力，甚至对主业务产生影响。为了实现读取能力的弹性扩展，分担数据库压力，您可以在某个区域中创建一个或多个只读实例，利用只读实例满足大量的数据库读取需求，以此增加应用的吞吐量。主实例和只读实例之间的数据同步不受网络延时的影响，只读实例跟主实例在同一区域，但可以在不同的可用区。以下教程将指导您如何通过Terraform脚本创建一个RDS只读实例。

相关资源

[huaweicloud_rds_read_replica_instance](#)

操作步骤

步骤1 创建MySQL数据库，请参考[8.1 创建MySQL数据库](#)。

步骤2 以MySQL为例，演示为RDS创建只读副本。

```
data "huaweicloud_availability_zones" "myaz" {}

resource "huaweicloud_rds_read_replica_instance" "myreplica" {
  name           = "myreplica"
  flavor         = "rds.mysql.c2.large.rr"
  primary_instance_id = huaweicloud_rds_instance.myinstance.id
  availability_zone = data.huaweicloud_availability_zones.myaz.names[1]

  volume {
    type = "ULTRAHIGH"
  }
  tags = {
    type = "readonly"
  }
}
```

- 1. 运行 terraform plan 查看资源。
- 2. 确认资源无误后，运行 terraform apply 开始创建。
- 3. 运行 terraform show 可以查看已创建的RDS只读实例信息。

----结束

表 8-3 参数说明

资源名称	参数名称	参数说明
huaweicloud_rds_read_replica_instance	name	（必填）只读实例名称 <ul style="list-style-type: none">约束：4~64个字符之间，必须以字母开头，区分大小写，可以包含字母、数字、中划线或者下划线，不能包含其他的特殊字符
	flavor	（必填）数据库只读实例规格，本教程使用 "rds.mysql.c2.large.rr"，实例规格信息也可以通过huaweicloud_rds_flavors查询。

资源名称		参数名称	参数说明
		primary_instance_id	(必填) 主实例ID
		availability_zone	(必填) 只读实例所在可用区，请参见 地区和终端节点 。
		tags	(可选) 实例标签。
		volume_type	(必填) 数据库只读实例磁盘类型 <ul style="list-style-type: none">取值范围： ULTRAHIGH，表示SSD； ULTRAHIGHPRO，表示SSD尊享版，仅支持超高性能型尊享版

样例代码

<https://github.com/huaweicloud/terraform-provider-huaweicloud/tree/master/examples/rds/read-replica>